Rules for integrands of the form $(e\,x)^m\left(a\,x^j+b\,x^k\right)^p(c+d\,x^n)^q$

1. $\int (e\,x)^m\left(a\,x^j+b\,x^k\right)^p(c+d\,x^n)^q\,dx$ when $p\notin\mathbb{Z}\land j\neq k\land\frac{j}{n}\in\mathbb{Z}\land\frac{k}{n}\in\mathbb{Z}\land\frac{m+1}{n}\in\mathbb{Z}\land n^2\neq 1$

   1: $\int x^m\left(a\,x^j+b\,x^k\right)^p(c+d\,x^n)^q\,dx$ when $p\notin\mathbb{Z}\land j\neq k\land\frac{j}{n}\in\mathbb{Z}\land\frac{k}{n}\in\mathbb{Z}\land\frac{m+1}{n}\in\mathbb{Z}\land n^2\neq 1$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n}\in\mathbb{Z}$, then $x^m F[x^n] == \frac{1}{n}\,\text{Subst}\left[x^{\frac{m+1}{n}-1}F[x],\,x,\,x^n\right]\partial_x x^n$

Rule: If $p\notin\mathbb{Z}\land j\neq k\land\frac{j}{n}\in\mathbb{Z}\land\frac{k}{n}\in\mathbb{Z}\land\frac{m+1}{n}\in\mathbb{Z}\land n^2\neq 1$, then

$$\int x^m\left(a\,x^j+b\,x^k\right)^p(c+d\,x^n)^q\,dx\longrightarrow\frac{1}{n}\,\text{Subst}\left[\int x^{\frac{m+1}{n}-1}\left(a\,x^{j/n}+b\,x^{k/n}\right)^p(c+d\,x)^q\,dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*(a_.*x_^j_+b_.*x_^k_.)^p_*(c_+d_.*x_^n_)^q_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a*x^Simplify[j/n]+b*x^Simplify[k/n])^p*(c+d*x)^q,x],x,x^n] /;
FreeQ[{a,b,c,d,j,k,m,n,p,q},x] && Not[IntegerQ[p]] && NeQ[k,j] && IntegerQ[Simplify[j/n]] && IntegerQ[Simplify[k/n]] &&
  IntegerQ[Simplify[(m+1)/n]] && NeQ[n^2,1]
```

2: $\int (e\,x)^m \left(a\,x^j+b\,x^k\right)^p \left(c+d\,x^n\right)^q dx$ when $p\notin\mathbb{Z}\ \land\ j\neq k\ \land\ \frac{j}{n}\in\mathbb{Z}\ \land\ \frac{k}{n}\in\mathbb{Z}\ \land\ \frac{m+1}{n}\in\mathbb{Z}\ \land\ n^2\neq 1$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(e\,x)^m}{x^m} = 0$

Basis: $\frac{(e\,x)^m}{x^m} = \frac{e^{\text{IntPart}[m]}\,(e\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule: If $p\notin\mathbb{Z}\ \land\ j\neq k\ \land\ \frac{j}{n}\in\mathbb{Z}\ \land\ \frac{k}{n}\in\mathbb{Z}\ \land\ \frac{m+1}{n}\in\mathbb{Z}\ \land\ n^2\neq 1$, then

$$\int (e\,x)^m \left(a\,x^j+b\,x^k\right)^p \left(c+d\,x^n\right)^q dx \longrightarrow \frac{e^{\text{IntPart}[m]}\,(e\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int x^m \left(a\,x^j+b\,x^k\right)^p \left(c+d\,x^n\right)^q dx$$

–

Program code:

```
Int[(e_*x_)^m_.*(a_.*x_^j_+b_.*x_^k_.)^p_*(c_+d_.*x_^n_.)^q_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a*x^j+b*x^k)^p*(c+d*x^n)^q,x] /;
FreeQ[{a,b,c,d,e,j,k,m,n,p,q},x] && Not[IntegerQ[p]] && NeQ[k,j] && IntegerQ[Simplify[j/n]] && IntegerQ[Simplify[k/n]] &&
  IntegerQ[Simplify[(m+1)/n]] && NeQ[n^2,1]
```

2. $\int (e\,x)^m \left(a\,x^j + b\,x^{j+n}\right)^p \left(c + d\,x^n\right) dx$ when $p \notin \mathbb{Z} \land b\,c - a\,d \neq 0$

1: $\int (e\,x)^m \left(a\,x^j + b\,x^{j+n}\right)^p \left(c + d\,x^n\right) dx$ when $p \notin \mathbb{Z} \land b\,c - a\,d \neq 0 \land a\,d\,(m + j\,p + 1) - b\,c\,(m + n + p\,(j + n) + 1) == 0 \land (e > 0 \lor j \in \mathbb{Z}) \land m + j\,p + 1 \neq 0$

Derivation: Trinomial recurrence 3b with $c = 0$ and $a\,d\,(m + j\,p + 1) - b\,c\,(m + n + p\,(j + n) + 1) == 0$

Rule: If
$p \notin \mathbb{Z} \land b\,c - a\,d \neq 0 \land a\,d\,(m + j\,p + 1) - b\,c\,(m + n + p\,(j + n) + 1) == 0 \land (e > 0 \lor j \in \mathbb{Z}) \land m + j\,p + 1 \neq 0$,
then

$$\int (e\,x)^m \left(a\,x^j + b\,x^{j+n}\right)^p \left(c + d\,x^n\right) dx \longrightarrow \frac{c\,e^{j-1}\,(e\,x)^{m-j+1} \left(a\,x^j + b\,x^{j+n}\right)^{p+1}}{a\,(m + j\,p + 1)}$$

Program code:

```
Int[(e_.*x_)^m_.*(a_.*x_^j_.+b_.*x_^jn_.)^p_*(c_+d_.*x_^n_.),x_Symbol] :=
  c*e^(j-1)*(e*x)^(m-j+1)*(a*x^j+b*x^(j+n))^(p+1)/(a*(m+j*p+1)) /;
FreeQ[{a,b,c,d,e,j,m,n,p},x] && EqQ[jn,j+n] && Not[IntegerQ[p]] && NeQ[b*c-a*d,0] && EqQ[a*d*(m+j*p+1)-b*c*(m+n+p*(j+n)+1),0] &&
  (GtQ[e,0] || IntegersQ[j]) && NeQ[m+j*p+1,0]
```

2: $\int (e\,x)^m \left(a\,x^j + b\,x^{j+n}\right)^p \left(c + d\,x^n\right) dx$ when $p \notin \mathbb{Z} \land b\,c - a\,d \neq 0 \land p < -1 \land 0 < j \leq m \land (e > 0 \lor j \in \mathbb{Z})$

Derivation: Trinomial recurrence 2b with $c = 0$

Rule: If $p \notin \mathbb{Z} \land b\,c - a\,d \neq 0 \land p < -1 \land 0 < j \leq m \land (e > 0 \lor j \in \mathbb{Z})$, then

$$\int (e\,x)^m \left(a\,x^j + b\,x^{j+n}\right)^p \left(c + d\,x^n\right) dx \longrightarrow$$

$$-\frac{e^{j-1}\,(b\,c - a\,d)\,(e\,x)^{m-j+1}\left(a\,x^j + b\,x^{j+n}\right)^{p+1}}{a\,b\,n\,(p+1)} - \frac{e^j\left(a\,d\,(m + j\,p + 1) - b\,c\,(m + n + p\,(j + n) + 1)\right)}{a\,b\,n\,(p+1)} \int (e\,x)^{m-j}\left(a\,x^j + b\,x^{j+n}\right)^{p+1} dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_.*x_^j_.+b_.*x_^jn_.)^p_*(c_+d_.*x_^n_.),x_Symbol] :=
  -e^(j-1)*(b*c-a*d)*(e*x)^(m-j+1)*(a*x^j+b*x^(j+n))^(p+1)/(a*b*n*(p+1)) -
  e^j*(a*d*(m+j*p+1)-b*c*(m+n+p*(j+n)+1))/(a*b*n*(p+1))*Int[(e*x)^(m-j)*(a*x^j+b*x^(j+n))^(p+1),x] /;
FreeQ[{a,b,c,d,e,j,m,n},x] && EqQ[jn,j+n] && Not[IntegerQ[p]] && NeQ[b*c-a*d,0] && LtQ[p,-1] && GtQ[j,0] && LeQ[j,m] &&
  (GtQ[e,0] || IntegerQ[j])
```

3: $\int (e\,x)^m \left(a\,x^j + b\,x^{j+n}\right)^p \left(c + d\,x^n\right) dx$ when $p \notin \mathbb{Z} \wedge b\,c - a\,d \neq 0 \wedge m < -1 \wedge n > 0 \wedge \left(e > 0 \vee (j \mid n) \in \mathbb{Z}\right)$

Derivation: Trinomial recurrence 3b with $c = 0$

Rule: If $p \notin \mathbb{Z} \wedge b\,c - a\,d \neq 0 \wedge m < -1 \wedge n > 0 \wedge \left(e > 0 \vee (j \mid n) \in \mathbb{Z}\right)$, then

$$\int (e\,x)^m \left(a\,x^j + b\,x^{j+n}\right)^p \left(c + d\,x^n\right) dx \longrightarrow$$

$$\frac{c\,e^{j-1}\,(e\,x)^{m-j+1} \left(a\,x^j + b\,x^{j+n}\right)^{p+1}}{a\,(m + j\,p + 1)} + \frac{a\,d\,(m + j\,p + 1) - b\,c\,(m + n + p\,(j + n) + 1)}{a\,e^n\,(m + j\,p + 1)} \int (e\,x)^{m+n} \left(a\,x^j + b\,x^{j+n}\right)^p dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_.*x_^j_.+b_.*x_^jn_.)^p_*(c_+d_.*x_^n_.),x_Symbol] :=
  c*e^(j-1)*(e*x)^(m-j+1)*(a*x^j+b*x^(j+n))^(p+1)/(a*(m+j*p+1)) +
  (a*d*(m+j*p+1)-b*c*(m+n+p*(j+n)+1))/(a*e^n*(m+j*p+1))*Int[(e*x)^(m+n)*(a*x^j+b*x^(j+n))^p,x] /;
FreeQ[{a,b,c,d,e,j,p},x] && EqQ[jn,j+n] && Not[IntegerQ[p]] && NeQ[b*c-a*d,0] && GtQ[n,0] &&
  (LtQ[m+j*p,-1] || IntegersQ[m-1/2,p-1/2] && LtQ[p,0] && LtQ[m,-n*p-1]) &&
  (GtQ[e,0] || IntegersQ[j,n]) && NeQ[m+j*p+1,0] && NeQ[m-n+j*p+1,0]
```

4: $\int (e\,x)^m \left(a\,x^j + b\,x^{j+n}\right)^p \left(c + d\,x^n\right) dx$ when $p \notin \mathbb{Z} \land b\,c - a\,d \neq 0 \land m + n + p\,(j + n) + 1 \neq 0 \land (e > 0 \lor j \in \mathbb{Z})$

Derivation: Trinomial recurrence 2b with $c = 0$ composed with binomial recurrence 1b

Rule: If $p \notin \mathbb{Z} \land b\,c - a\,d \neq 0 \land m + n + p\,(j + n) + 1 \neq 0 \land (e > 0 \lor j \in \mathbb{Z})$, then

$$\int (e\,x)^m \left(a\,x^j + b\,x^{j+n}\right)^p \left(c + d\,x^n\right) dx \longrightarrow$$
$$\frac{d\,e^{j-1}\,(e\,x)^{m-j+1} \left(a\,x^j + b\,x^{j+n}\right)^{p+1}}{b\,(m + n + p\,(j + n) + 1)} - \frac{a\,d\,(m + j\,p + 1) - b\,c\,(m + n + p\,(j + n) + 1)}{b\,(m + n + p\,(j + n) + 1)} \int (e\,x)^m \left(a\,x^j + b\,x^{j+n}\right)^p dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_.*x_^j_.+b_.*x_^jn_.)^p_*(c_+d_.*x_^n_.),x_Symbol] :=
  d*e^(j-1)*(e*x)^(m-j+1)*(a*x^j+b*x^(j+n))^(p+1)/(b*(m+n+p*(j+n)+1)) -
  (a*d*(m+j*p+1)-b*c*(m+n+p*(j+n)+1))/(b*(m+n+p*(j+n)+1))*Int[(e*x)^m*(a*x^j+b*x^(j+n))^p,x] /;
FreeQ[{a,b,c,d,e,j,m,n,p},x] && EqQ[jn,j+n] && Not[IntegerQ[p]] && NeQ[b*c-a*d,0] && NeQ[m+n+p*(j+n)+1,0] && (GtQ[e,0] || IntegerQ[j])
```

3. $\int (e\,x)^m \left(a\,x^j+b\,x^k\right)^p \left(c+d\,x^n\right)^q\,dx$ when $p\notin\mathbb{Z}\,\land\,j\neq k\,\land\,\frac{j}{n}\in\mathbb{Z}\,\land\,\frac{k}{n}\in\mathbb{Z}\,\land\,\frac{n}{m+1}\in\mathbb{Z}$

1: $\int x^m \left(a\,x^j+b\,x^k\right)^p \left(c+d\,x^n\right)^q\,dx$ when $p\notin\mathbb{Z}\,\land\,j\neq k\,\land\,\frac{j}{n}\in\mathbb{Z}\,\land\,\frac{k}{n}\in\mathbb{Z}\,\land\,\frac{n}{m+1}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{n}{m+1}\in\mathbb{Z}$, then $x^m\,F[x^n]=\frac{1}{m+1}\,\mathrm{Subst}\left[F\left[x^{\frac{n}{m+1}}\right],\,x,\,x^{m+1}\right]\partial_x x^{m+1}$

Rule: If $p\notin\mathbb{Z}\,\land\,j\neq k\,\land\,\frac{j}{n}\in\mathbb{Z}\,\land\,\frac{k}{n}\in\mathbb{Z}\,\land\,\frac{n}{m+1}\in\mathbb{Z}$

$$\int x^m \left(a\,x^j+b\,x^k\right)^p \left(c+d\,x^n\right)^q\,dx \longrightarrow \frac{1}{m+1}\,\mathrm{Subst}\left[\int\left(a\,x^{\frac{j}{m+1}}+b\,x^{\frac{k}{m+1}}\right)^p \left(c+d\,x^{\frac{n}{m+1}}\right)^q dx,\,x,\,x^{m+1}\right]$$

–

Program code:

```
Int[x_^m_.*(a_.*x_^j_+b_.*x_^k_.)^p_*(c_+d_.*x_^n_.)^q_.,x_Symbol] :=
  1/(m+1)*Subst[Int[(a*x^Simplify[j/(m+1)]+b*x^Simplify[k/(m+1)])^p*(c+d*x^Simplify[n/(m+1)])^q,x],x,x^(m+1)] /;
FreeQ[{a,b,c,d,j,k,m,n,p,q},x] && Not[IntegerQ[p]] && NeQ[k,j] && IntegerQ[Simplify[j/n]] && IntegerQ[Simplify[k/n]] &&
  NeQ[m,-1] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

2: $\int (e\,x)^m \left(a\,x^j+b\,x^k\right)^p \left(c+d\,x^n\right)^q\,dx$ when $p\notin\mathbb{Z}\,\land\,j\neq k\,\land\,\frac{j}{n}\in\mathbb{Z}\,\land\,\frac{k}{n}\in\mathbb{Z}\,\land\,\frac{n}{m+1}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(e\,x)^m}{x^m} = 0$

Basis: $\frac{(e\,x)^m}{x^m} = \frac{e^{\text{IntPart}[m]}\,(e\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule: If $p\notin\mathbb{Z}\,\land\,j\neq k\,\land\,\frac{j}{n}\in\mathbb{Z}\,\land\,\frac{k}{n}\in\mathbb{Z}\,\land\,\frac{n}{m+1}\in\mathbb{Z}$, then

$$\int (e\,x)^m \left(a\,x^j+b\,x^k\right)^p \left(c+d\,x^n\right)^q\,dx \longrightarrow \frac{e^{\text{IntPart}[m]}\,(e\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m \left(a\,x^j+b\,x^k\right)^p \left(c+d\,x^n\right)^q\,dx$$

–

Program code:

```
Int[(e_.*x_)^m_.*(a_.*x_^j_+b_.*x_^k_.)^p_*(c_+d_.*x_^n_.)^q_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a*x^j+b*x^k)^p*(c+d*x^n)^q,x] /;
FreeQ[{a,b,c,d,e,j,k,m,n,p,q},x] && Not[IntegerQ[p]] && NeQ[k,j] && IntegerQ[Simplify[j/n]] && IntegerQ[Simplify[k/n]] &&
  NeQ[m,-1] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

4: $\int (e\,x)^m \left(a\,x^j+b\,x^{j+n}\right)^p \left(c+d\,x^n\right)^q\,dx$ when $p\notin\mathbb{Z}\,\land\,b\,c-a\,d\neq 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(e\,x)^m \left(a\,x^j+b\,x^{j+n}\right)^p}{x^{m+j\,p}\,\left(a+b\,x^n\right)^p} = 0$

Basis: $\frac{(e\,x)^m}{x^m} = \frac{e^{\text{IntPart}[m]}\,(e\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Basis: $\frac{\left(a\,x^j+b\,x^{j+n}\right)^p}{x^{j\,p}\,\left(a+b\,x^n\right)^p} = \frac{\left(a\,x^j+b\,x^{j+n}\right)^{\text{FracPart}[p]}}{x^{j\,\text{FracPart}[p]}\,\left(a+b\,x^n\right)^{\text{FracPart}[p]}}$

–

Rule: If $p\notin\mathbb{Z}\,\land\,b\,c-a\,d\neq 0$, then

$$\int (e\,x)^m \left(a\,x^j+b\,x^{j+n}\right)^p \left(c+d\,x^n\right)^q dx \longrightarrow \frac{(e\,x)^m \left(a\,x^j+b\,x^{j+n}\right)^p}{x^{m+j\,p} \left(a+b\,x^n\right)^p} \int x^{m+j\,p} \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q dx$$

$$\longrightarrow \frac{e^{\text{IntPart}[m]} \,(e\,x)^{\text{FracPart}[m]} \left(a\,x^j+b\,x^{j+n}\right)^{\text{FracPart}[p]}}{x^{\text{FracPart}[m]+j\,\text{FracPart}[p]} \left(a+b\,x^n\right)^{\text{FracPart}[p]}} \int x^{m+j\,p} \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_.*x_^j_.+b_.*x_^jn_.)^p_*(c_+d_.*x_^n_.)^q_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]*(a*x^j+b*x^(j+n))^FracPart[p]/
    (x^(FracPart[m]+j*FracPart[p])*(a+b*x^n)^FracPart[p])*
    Int[x^(m+j*p)*(a+b*x^n)^p*(c+d*x^n)^q,x] /;
FreeQ[{a,b,c,d,e,j,m,n,p,q},x] && EqQ[jn,j+n] && Not[IntegerQ[p]] && NeQ[b*c-a*d,0] && Not[EqQ[n,1] && EqQ[j,1]]
```